
1. Presentación de ProScreenerTM 

ProScreenerTM es una sofisticada herramienta de rastreo que le permitirá rastrear 

mercados enteros:  

 Buscando valores que cumplan una o varias condiciones definidas por usted 

mismo(a)  

 En una o múltiples unidades de tiempo (p.ej.: 1 minuto y 1 hora)  

 En tiempo real o al cierre de cada vela, con una precisión de tick por tick  

ProScreenerTM integra el lenguaje de programación ProBuilderTM (que le aconsejamos 

conocer, leyendo la Guía de programación ProBuilderTM antes de leer el de 

ProScreenerTM) con extensiones que se aplican exclusivamente a las condiciones de 

filtrado. Los rastreos en tiempo real o al cierre de la vela pueden aplicarse en las 

siguientes unidades de tiempo:  

 1 minuto 

 2 minutos 

 3 minutos 

 5 minutos 

 10 minutos 

 15 minutos 

 30 minutos 

 1 hora 

 2 horas 

 3 horas 

 4 horas 

 Diario 

 Semanal 

 

Los resultados de un escaneo ProScreenerTM se muestran en una lista de resultados con 

los 30 mejores valores según el criterio de ordenación definido.  

Este documento debe considerarse como una extensión de la Guía de uso ProBuilderTM 
aunque también puede leerse de manera independiente. Mediante un enfoque teórico y 
ejemplos concretos, podrá aprender las bases de la programación de códigos que le 
permitirán definir filtros aplicables sobre cualquier mercado incluido en la cobertura de 
Broker Plus.  

En el final de esta guía de uso, encontrará un glosario que le proporciona un listado 
completo de todos los comandos que pueden utilizarse.  

2. Nociones fundamentales 

Acceder a ProScreener 

Puede acceder al módulo ProScreenerTM pulsando en 'Mostrar' -> 'ProScreener', o con 

las teclas Ctrl+E:  

 

Esto hará aparecer la ventana ProScreener, donde puede seleccionar un código ya 

existente o programar un código nuevo. Para ello, acceda a la ventana de programación 

pulsando 'Gestión ProScreeners' -> 'Nuevo ProScreener'. Encontrará dos opciones:  

 La 'Creación asistida', que le permite definir condiciones de búsqueda sin 

necesidad de teclear código.  

 La 'Creación programando', que le permite definir condiciones más sofisticadas y 

sus parámetros de aplicación mediante código.  

 

La ventana 'Creación programando' contiene 4 secciones:  

1) Zona de programación del ProScreenerTM 

2) Selección del mercado en el que se efectuará el rastreo  

3) Período considerado  

4) Criterio de clasificación de resultados  

En la primera sección podrá:  

 Programar directamente un ProScreenerTM en la zona de texto  

 Utilizar la función de ayuda 'Insertar función", que permite hallar en una nueva 

ventana la librería de funciones clasificadas en 7 categorías.  

Empezaremos utilizando la opción 'Insertar función' pulsando en el botón 

correspondiente. 

Tomemos como ejemplo la instrucción 'SCREENER', que permite buscar condiciones. 

Esta instrucción se encuentra en la categoría 'Comandos ProScreener'. 

Pulse en la palabra 'SCREENER' -> 'Añadir': el comando aparecerá en la zona de 

programación. 

El comando 'SCREENER' le permite definir la(s) condición(es) de búsqueda. 

La última línea de cualquier ProScreenerTM deberá comenzar por el comando 

'SCREENER'. 

 

Supongamos que deseamos buscar todos los valores del IBEX 35 en los que la apertura 
de la vela (período) actual sea mayor que el cierre de la vela precedente. El código 
correspondiente sería:  

c1 = (Open > Close[1])  
SCREENER[c1]  

 

Una vez definido el código, vamos a elegir la segunda sección, una lista predefinida o 
una personalizada en la que efectuar la búsqueda.  

Nota: La búsqueda se realiza exclusivamente en valores de un único mercado. Para 
aplicar un ProScreenerTM a una lista personalizada, deberá asegurarse de que todos los 
valores que la componen pertenezcan al mismo mercado.  

La tercera sección le permite definir el período utilizado durante el rastreo del 
ProScreenerTM. Dado que la cotización puede verse en velas de períodos distintos, una 
configuración que se dé en velas de una unidad de tiempo determinada podría no darse 
en otra unidad de tiempo distinta.  

La cuarta sección controla la clasificación de los resultados en función del grado de 
intensidad del cumplimiento de los criterios definidos en el código (ver la sección: 
"Instrucciones de búsqueda y filtrado"). Si las condiciones son poco restrictivas, los 
resultados pueden ser muy amplios (llegando a varios millares en el caso de mercados 
con muchos valores cotizados). Por ello, es conveniente filtrar los resultados según uno 
de estos criterios:  

 Los 30 valores que cumplen los criterios con mayor intensidad  

 Los 30 valores que cumplen los criterios con menor intensidad  

Ejemplo: supongamos que su ProScreenerTM rastrea todo el Mercado Continuo buscando 
en velas diarias todos los valores que tengan un volumen diario superior a 20.000. 
Cabe esperar que el número de resultados de tal búsqueda será superior a la treintena. 
Así, parametrizar el criterio de filtraje por ejemplo en el % de variación de los valores le 
permitirá visualizar los resultados con las variaciones más amplias o las menos amplias.  

Tras definir estos 4 parámetros, pulse 'Validar programa' para lanzar la búsqueda. Si 
ésta proporciona resultados, se mostrarán en una lista similar a ésta:  

Las ventanas Top Variación  

Además del módulo ProScreenerTM, Broker Plus le ofrece versiones predefinidas con un 
abanico de búsquedas en las ventanas Top Variación. Estas ventanas rastrean los 
mercados con criterios predefinidos (estos criterios son completamente personalizables 
en el módulo ProScreenerTM). Ambas ventanas son idénticas y le permitirán lanzar 
búsquedas simultáneas en el mismo o en distintos mercados (ejemplo: variación al alza 
en una ventana y a la baja en la otra).  

Los criterios de búsqueda propuestos en estas ventanas se agrupan en 4 categorías, 

capaces de detectar:  

 Variaciones y gaps  

 Las mayores variaciones en la preapertura del mercado 

 Las principales configuraciones de velas japonesas 

 Los valores de Spread y Volumen en el libro de órdenes de compraventa  

3. Programar con ProScreenerTM 

En este capítulo le mostramos los 5 comandos ProBuilderTM específicos al módulo 

ProScreenerTM. Los encontrará también en la biblioteca de funciones, en la sección 

'Comandos ProScreener'. Vamos a revisar los siguientes:  

 Instrucciones de búsqueda y filtrado  

 Anticipar el volumen 

 Búsqueda en períodos múltiples 

 Búsqueda de múltiples valores  

 

Instrucciones de búsqueda y filtrado  

El comando 'SCREENER' lanza la búsqueda. Es el equivalente a 'RETURN' en el 

lenguaje ProBuilderTM; al igual que éste, puede acompañarse de muchas funciones 

complementarias que vamos a estudiar a continuación.  

'SCREENER' se utiliza de la siguiente manera:  

SCREENER[Condición] 

Tomemos un ejemplo:  

Vamos a buscar valores de un mercado en los que el precio al cierre sea estrictamente 
menor a la banda Bollinger inferior. Las bandas de Bollinger se calculan sobre 10 
periodos y se aplican al precio al cierre.  

c1 = (Close < BollingerDown[10](Close))  
SCREENER[c1]  

 

Podemos ampliar la búsqueda a condiciones múltiples, y establecer que éstas se 
cumplan simultáneamente o alternativamente. En tal caso, la sintaxis de ProScreenerTM 
sería:  

SCREENER[Condición1 AND Condición2] 
 

En el caso de que todas las condiciones deban cumplirse simultáneamente, o bien  

SCREENER[Condición1 OR Condición2] 

Si queremos establecer que al menos una de las condiciones deba cumplirse.  

Veamos un ejemplo:  

Busquemos valores en los que el precio sea mayor que el canal de Bollinger y que 
tengan una tendencia al alza.  

REM Cierre por encima de la Boll sup  
Condition1 = (Close > Bollingerup[20](Close))  
REM Cierre > a la Apertura  
Condition2 = Close > Open  
REM MM7 > MM23  
Condition3 = (Average[7](Close) > Average[23](Close))  
SCREENER[Condition1 AND Condition2 AND Condition3]  

 

También podemos definir una constante o un indicador (predefinido o personalizado) 
que ejercerá simultáneamente de criterio de filtrado y de clasificación de resultados. 
Así,  

 Si la búsqueda proporciona más de 30 resultados, el indicador filtra los valores que 
serán visualizados en la lista. En este caso, puede utilizar la cuarta sección de la 
ventana de programación del ProScreenerTM para definir el orden de aparición de 
estos resultados (ver la descripción en la p. 7).  

 Si el número de resultados es igual o menor a 30, el indicador los ordena según la 
base de este nuevo indicador. 

La sintaxis para insertar una constante es:  

SCREENER[c1](TypicalPrice) 

En el caso de un indicador predefinido, deberemos comenzar por almacenarlo en una 
variable (que en este caso, llamaremos 'Criterio'):  

Criterio = RSI[14](Close) 

SCREENER[c1](Criterio) 

Si utilizamos un indicador personalizado, utilizaremos la instrucción 'CALL' (cuya 
definición se encuentra en la Guía de uso ProBuilderTM).  

MiRSI = CALL "RSI" [14] 

SCREENER[c1](MiRSI) 

Ejemplo:  

Busquemos todos los valores de un mercado cuyo volumen de negociación sea mayor 
que 50.000. Crearemos un indicador que nos permita evaluar si el RSI se encuentra en 
situación de sobrecompra o sobreventa. Mostraremos los resultados del ProScreenerTM 
ordenados a partir de dicho indicador.  

c1 = Volume > 50000  

IF RSI[20](Close) > 70 THEN  

Criterio = 1  

ELSIF RSI[20](Close) < 30 THEN  

Criterio = -1  

ENDIF  

SCREENER[c1](Criterio AS "sobrecompra")  

También podemos omitir el uso de paréntesis introduciendo el comando 'SORT BY':  

SCREENER[c1] SORT BY TypicalPrice 

Si queremos atribuir un nombre personalizado esta columna, deberemos utilizar la 
función 'AS':  

SCREENER[c1](TypicalPrice AS "PrecioTipico") 

 

Anticipar el volumen  

El comando 'EstimatedVolume' proporciona una estimación linear del volumen de la 
barra actual. Para ser exactos, se calcula el volumen estimado a partir de la siguiente 
fórmula:  

Volumen_estimado = Volumen * Coeficiente_multiplicador  

ó  

Coeficiente_multiplicador = duración de la vista / tiempo transcurrido 
desde la creación de la vela  

Este comando es especialmente interesante para comparar el volumen estimado con el 
volumen real.  

Veamos un ejemplo:  

En vista 10 minutos, si suponemos que la última vela se está formando desde hace 1 
minuto, el volumen estimado será 10 veces el volumen actual.  
Calculemos la relación entre el volumen anticipado del día respecto al volumen de la 
víspera (para obtener resultados representativos, es aconsejable utilizar un período 
intradiario):  

 

REM Evaluación del volumen del día  

Vol0 = EstimatedVolume  

REM Hallar el volumen de la víspera  

Vol1 = Volume[1]  

REM Criterio de clasificación: relación entre el volumen anticipado del 
día respecto al volumen de la víspera  

SCREENER (Vol0 / Vol1 AS "Volumen")  

Búsqueda en períodos múltiples  

Puede definir una búsqueda que verifique los criterios en múltiples unidades de tiempo. 
Esto permite comprobar que una condición se satisfaga, por ejemplo a corto o medio 
plazo. El comando adaptado se denomina 'TIMEFRAME', y su sintaxis es la siguiente:  

TIMEFRAME (unidad de tiempo del período) 

Las unidades de tiempo disponibles se traducen en código por: 

Interpretación Código  

1 minuto 1 minute 

2 minutos 2 minutes 

3 minutos 3 minutes 

5 minutos 5 minutes 

10 minutos 10 minutes 

15 minutos 15 minutes 

30 minutos 30 minutes 

1 hora 1 hour 

2 horas 2 hours 

3 horas 3 hours 

4 horas 4 hours 

Diario daily 

Semanal weekly 
 

Las instrucciones que siguen a la línea 'TIMEFRAME' se buscarán únicamente en el 
período indicado. Es posible crear varias líneas 'TIMEFRAME' en un mismo código, 
para efectuar una búsqueda en períodos múltiples.  

Tomemos un ejemplo: Busquemos todos los valores del IBEX 35 que cumplan los 
siguientes criterios:  

 En vista semanal, el indicador Williams %R de 14 períodos debe tener un valor entre 
0 y -20.  

 En vista de 30 minutos, la media móvil exponencial de 20 períodos cruza al alza la de 

12 períodos.  

Podríamos detectar los valores situados en zona de sobrecompra en períodos semanales 
y que presenten a corto plazo una tendencia bajista.  

 

TIMEFRAME(weekly)  

Condition1 = Williams[14](Close) < 0 AND Williams[14](Close) > -20  

TIMEFRAME(30 minutes)  

Condition2 = ExponentialAverage[20](Close) CROSSES OVER 
ExponentialAverage[12](Close)  

SCREENER[Condicion1 AND Condicion2]  

 

Búsqueda de múltiples valores  

Hemos constatado que con ProScreenerTM se debe especificar un mercado o una lista de 
valores sobre el que efectuar la búsqueda.  

La instrucción 'EQUITYFRAME' permite relacionar el criterio con un valor específico 
perteneciente al mismo mercado en el que se realiza la búsqueda.  

Se utiliza de esta manera:  

EQUITYFRAME("nombre del mercado" , "ticker") 

El comando compara los resultados con un valor concreto y permite la creación de un 
nuevo indicador que puede utilizarse como condición de ordenación. El uso optimizado 
de este comando requiere que se respeten estas condiciones:  

 Debe emplearse exclusivamente en mercados en los que se disponga de acceso a 

datos en tiempo real  

 No puede aplicarse a los índices de los mercados  

 El valor elegido debe formar parte de la misma lista de búsqueda (ver sección 2 de la 

ventana ProScreenerTM)  

 

Así, para visualizar una treintena de valores del mercado IBEX 35 seleccionados 
aleatoriamente y relacionados con el precio de cierre actual del valor TELEFONICA 
(ticker 'TEF'), el código será:  

EQUITYFRAME("Mercado Continuo" , "TEF")  

MiCierre = Close  

SCREENER(MiCierre)  

Al igual que en el caso del comando 'TIMEFRAME', el código que sigue a 
'EQUITYFRAME' se aplica al valor especificado por este último. Para volver a los datos 
del valor seleccionado en la interfaz de ProScreenerTM, podemos escribir:  

EQUITYFRAME(default) 

Las dos funciones descritas son complementarias, tal y como se ilustra en el siguiente 

ejemplo:  

Si deseamos visualizar el indicador que muestra la diferencia de precio al cierre de la 

vela actual del valor TELEFONICA (ticker 'TEF') respecto al precio de cada uno de treinta  

valores del IBEX 35 seleccionados al azar, el código sería el siguiente:  

 

vREM Comencemos por recuperar las informaciones a extraer del 
valor del IBEX 35  

EQUITYFRAME("Mercado Continuo" , "TEF")  

MiCierre = Close  

REM Volvemos a los valores del mercado seleccionados en la lista 
desplegable "Selección de la lista"  

EQUITYFRAME(default)  

REM Recuperamos las informaciones deseadas  

ValCierre = MiCierre -Close  

REM Lanzamos la visualización del resultado  

SCREENER(ValCierre AS "MiIndicador")  

 

Ejemplo:  

Vamos a representar visualmente la correlación entre un valor y su mercado, o entre un 
índice y su mercado. Calculamos el ratio de dos valores seleccionados 
('EQUITYFRAME') y acto seguido, la diferencia de este ratio respecto a la víspera.  

TIMEFRAME(Daily) 

ValCierre = Close 

EQUITYFRAME("NASDAQ" , "AMZN") 

IndCierre = Close 

EQUITYFRAME(default) 

Ratio = (ValCierre / IndCierre) * 100 

FuerzaRelativa = (Ratio -Ratio[1]) * 100 

SCREENER(FuerzaRelativa AS "Fuerza Relativa") 

4. Aplicaciones prácticas 

4.A) Ejemplos simples 

4.A.1) RSI 1hora: sobreventa  

El RSI es un indicador de sobrecompra / sobreventa que puede anticipar un cambio de 
tendencia. En este ejemplo, vamos a intentar detectar valores que se encuentren en 
situación de sobreventa.  

Definiremos la situación de sobreventa como RSI < 30. Cuanto más se acerque a cero 
el RSI, la situación de sobreventa será más acentuada. Construiremos un ProScreenerTM 
que nos busque los valores cuyo RSI sea menor que 30.  

Ejemplo RSI en sobreventa:  

REM Cálculo del RSI de 14 periodos  

MiRSI = RSI[14]  

REM Filtro: RSI < 30  

Filtro = MiRSI < 30  

SCREENER[Filtro](MiRSI AS "RSI")  

 

4.A.2) RSI 1hora: sobrecompra  

Busquemos ahora valores en situación de sobrecompra, definida como RSI > 70. 
Cuanto más se acerque al RSI a 100, la situación de sobrecompra será más acentuada. 
Construiremos un ProScreenerTM que nos busque los valores cuyo RSI se acerque a 100.  

Ejemplo RSI en sobrecompra:  

REM Cálculo del RSI de 14 periodos  

MiRSI = RSI[14]  

REM Filtro: RSI > 70  

Filtro = MiRSI > 70  

SCREENER[Filtro](MiRSI AS "RSI")  

4.A.3) Cruce alcista de medias móviles  

En este caso buscaremos resaltar una tendencia en la que una media móvil corta cruza 
una media móvil larga.  

Construiremos un ProScreenerTM que busque valores en los que la media móvil de 20 
períodos cruce al alza la media móvil de 50 períodos.  

Además, calcularemos el momento de la diferencia entre ambas medias móviles para 
evaluar la fuerza del cruce. Si este número es cercano a cero, el crecimiento será lento 
y las dos medias móviles deberían ser casi paralelas; la señal será poco significativa. A 
la inversa, cuanto más elevado sea este número, más 'dinámico' será el cruce.  

Un ejemplo típico lo constituye el de una media móvil plana a la que cruza otra media 
móvil con una fuerte tendencia al alza.  

Ejemplo: Cruce al alza de la media móvil a 50 períodos por la media móvil a 20 
períodos  

REM Cálculo de la media móvil simple de 20 periodos  

MM20 = Average[20]  

REM Cálculo de la media móvil simple de 50 periodos  

MM50 = Average[50]  

REM Evaluación de la velocidad relativa de la MM corta respecto a la larga  

Velocidad = Momentum(MM50 -MM20) * 100 / Close  

REM Selección del valor en el momento del cruce  

Filtro = MM20 Crosses Over MM50  

SCREENER[Filtro](Velocidad AS "Dinamica")  

4.A.4) Cruce bajista de medias móviles  

Construiremos un ProScreenerTM que busque valores en los que la media móvil de 20 
períodos cruce a la baja la media móvil de 50 períodos.  

El momento se define de modo análogo al descrito en el ejemplo precedente.  

Ejemplo: Cruce a la baja de la media móvil a 50 períodos por la media móvil a 20 
períodos  

REM Cálculo de la media móvil simple de 20 periodos  

MM20 = Average[20]  

REM Cálculo de la media móvil simple de 50 periodos  

MM50 = Average[50]  

REM Evaluación de la velocidad relativa de la MM corta respecto a la larga  

Velocidad = Momentum(MM20 -MM50) * 100 / Close  

REM Selección del valor en el momento del cruce  

Filtro = MM20 Crosses Under MM50  

SCREENER[Filtro](Velocidad AS "Dinamica")  

 

4.B) Ejemplos más sofisticados  

4.B.1)  RSI y cambio de tendencia alcista  

Presentaremos aquí un sistema que permita la detección en tiempo real de los valores 
más susceptibles de cambiar de tendencia, y creando además un movimiento de gran 
envergadura.  

Tradicionalmente, los analistas se interesan a los indicadores de sobrecalentamiento o a 
las figuras de cambio de tendencia aislándolas de su contexto (a menudo, por ausencia 
de herramientas que les permitiesen obrar de otro modo mejor).  

Con el módulo ProScreenerTM estas limitaciones técnicas desaparecen, y podemos 
construir un auténtico sistema de detección basado en el RSI. El punto de inicio es 
sencillo: buscamos un RSI en sobreventa con un cambio de tendencia al alza.  

Introduciremos un filtro que contenga estas dos condiciones. Pueden inscribirse en una 
sola línea de código:  

Filtro = RSI < 30 AND Momentum[1](RSI) > 0  

Una vez que tenemos los valores que puedan interesarnos, vamos a ordenarlos para 
mostrar aquellos que tengan un máximo interés. Para ello, consideraremos aquellos 
cuya caída haya sido más severa en relación a la volatilidad habitual del valor.  

Ejemplo: RSI y cambio de tendencia alcista  

REM Filtrado de valores con RSI en sobreventa y en cambio de 
tendencia al alza // Criterio de filtrado: RSI < 30 y creciente  

Filtro = RSI < 30 AND Momentum[1](RSI) > 0  

REM Determinar la "fuerza" de la tendencia bajista  

REM Hallar el punto máximo en las 20 últimas velas  

PuntoMaximo = Highest[20](High)  

REM Determinar la caída desde este punto  

Caida = PuntoMaximo -Close  

REM Determinar la volatilidad habitual del valor (mediana del True 
Range sobre 3 velas)  

Norma = Summation[3](TR) -Highest[3](TR) -Lowest[3](TR)  

Velocidad = Caida / Norma  

REM Mostrar resultados: bajadas relativas más importantes  

SCREENER[Filtro](Velocidad AS "Fuerza bajista")  

 

4.B.2) RSI y cambio de tendencia bajista  

El punto de inicio de este ProScreenerTM es la búsqueda de un RSI en sobrecompra que 
experimenta un cambio de tendencia a la baja.  

Al igual que en el caso anterior, crearemos un filtro en código ProBuilderTM:  

Filtro = RSI > 70 AND Momentum[1](RSI) < 0  

 

Al igual que con el ProScreenerTM previo, recuperaremos los valores en los que la subida 
sea más marcada en relación con la volatilidad habitual del valor.  

Ejemplo: RSI y cambio de tendencia bajista  

REM Filtrado de valores en los que el RSI está en sobrecompra y 
cambio de tendencia a la baja // Criterio de filtrado: RSI > 70 y 
decreciente  

Filtro = RSI > 70 AND Momentum[1](RSI) < 0  

REM Determinar la "fuerza" de la tendencia alcista // Hallar el punto 
mínimo en las 20 últimas velas previas  

PuntoMinimo = Lowest[20](Low)  

// Determinar la subida desde este punto  

Alza = Close -PuntoMinimo  

// Determinar la volatilidad habitual del valor (mediana del True Range 
sobre 3 velas)  

Norma = Summation[3](TR) -Highest[3](TR) -Lowest[3](TR)  

Velocidad = Alza / Norma  

REM Mostrar resultados: subidas relativas más importantes  

SCREENER[Filtro](Velocidad AS "Fuerza alcista")  

 

4.B.3) Vela envolvente alcista  

La vela envolvente alcista es una de las configuraciones de velas japonesas más 
conocidas, pero que evidentemente ha de ser tomada en cuenta con su contexto. 

Empecemos por definir una vela envolvente alcista: 

 Vela previa de color negro / rojo  

 Apertura de la vela actual por debajo del cuerpo de la vela previa  

 Cierre de la vela actual por encima del cuerpo de la vela previa  

El filtro que agrupa las tres condiciones será el siguiente:  

Filtro = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]  

En cuanto a la detección de la tendencia bajista, retomaremos el código de los ejemplos 
anteriores ajustando ligeramente el horizonte temporal. Con el RSI 14 era razonable 
interesarse a la caída de precios sobre las 20 velas precedentes. En el caso de las velas 
japonesas, la experiencia indica que basta con estudiar las 8 últimas velas (ya que se 
trata de configuraciones más reactivas y teóricamente limitadas a un máximo de 7 
velas).  

Ejemplo: Vela envolvente alcista  

REM Determinar la "fuerza" de la tendencia bajista  

// Buscar el máximo de las 8 velas previas  

Maximo = Highest[8](High)   

// Determinar la caída desde este punto  

Caida = Maximo -Close  

// Determinar la volatilidad habitual del valor (mediana del True Range 
sobre 3 velas)  

Norma = Summation[3](TR) -Highest[3](TR) -Lowest[3](TR)  

REM Criterio de filtrado: vela envolvente alcista  

Filtro = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]  

Velocidad = Caida / Norma  

REM Mostrar resultados: caídas relativas más importantes  

SCREENER[Filtro](Velocidad AS "Fuerza de la caida")  

 

4.B.4) Vela envolvente bajista  

Empecemos por definir una vela envolvente bajista:  

 Vela previa de color blanco / verde  

 Apertura de la vela actual por encima del cuerpo de la vela previa  

 Cierre de la vela actual por debajo del cuerpo de la vela previa  

El filtro que agrupa las tres condiciones será el siguiente:  

Filtro = Close[1] > Open[1] AND Open > Close[1] AND Close < Open[1]  

Para detectar la tendencia alcista, retomaremos el código del RSI con cambio de 
tendencia a la baja que habíamos creado previamente.  

Ejemplo: Vela envolvente bajista  

REM Determinar la "fuerza" de la tendencia alcista  

// Buscar el mínimo de las 8 velas previas  

Minimo = Lowest[8](Low)  

// Determinar el alza desde este punto  

Alza = Close -Minimo  

// Determinar la volatilidad habitual del valor (mediana del True Range 
sobre 3 velas)  

Norma = Summation[3](TR) -Highest[3](TR) -Lowest[3](TR)  

REM Criterio de filtrado: vela envolvente Filtro = Close[1] > Open[1] 
AND Open > Close[1] AND Close < Open[1]  

Velocidad = Alza / Norma  

REM Mostrar resultados: alzas relativas más importantes  

SCREENER[Filtro](Velocidad AS "Fuerza alcista")  

 

4.B.5) Triple pantalla alcista  

Este ProScreenerTM se compone de tres condiciones en varias unidades de tiempo:  

 Condición 1: MACD en vista semanal < 0  

 Condición 2: MACD creciente en vista semanal  

 Condición 3: Estocástico en vista diaria < 30  

Buscamos los valores más cercanos del nivel de compra STOP recomendado por el 
sistema (en este caso, se tratará del valor máximo de la víspera).  
 
El ProScreenerTM mostrará los valores situados por debajo de este nivel, así como 
aquellos que acaben de cruzarlo y que se mantienen en un margen cercano (no 
superior a un +5%).  

Los resultados respetarán las condiciones en las unidades de tiempo indicadas (vista 
semanal y diaria).  

Ejemplo: Triple pantalla alcista  

REM Condición 1: MACD semanal < 0 y creciente 

TIMEFRAME(weekly) 

MiMACD = MACD[12,26,9](Close) 

c1 = MiMACD < 0 AND MiMACD > MiMACD[1] 

REM Condición 2: Estocástico diario < 30 

TIMEFRAME(daily) 

MiSTO = Stochastic[14,3](Close) 

c2 = MiSTO < 30 

REM Nivel del stop de compra 

MiStop = High[1] 

REM Criterio: Posición de los precios respecto al STOP 

Criterio = (Close / MiStop -1) * 100 

REM Condición 3: los precios bajo el STOP o a una distancia inferior a 
un +5% 

c3 = Criterio < 5 

SCREENER[c1 AND c2 AND c3](Criterio) 

 

4.B.6) Triple pantalla bajista  

Este ProScreenerTM se compone de tres condiciones en varias unidades de tiempo:  

 Condición 1: MACD en vista semanal > 0  

 Condición 2: MACD decreciente en vista semanal  

 Condición 3: Estocástico en vista diaria > 70  

Buscamos los valores más cercanos del nivel e compra STOP recomendado por el 
sistema (en este caso, se tratará del valor mínimo de la víspera).  

El ProScreenerTM mostrará los valores situados por encima de este nivel, así como 
aquellos que acaben de cruzarlo y que se mantienen en un margen cercano (no 
superior a un -5%).  

Los resultados respetarán las condiciones en las unidades de tiempo indicadas (vista 
semanal y diaria).  

Ejemplo: Triple pantalla bajista  

REM Condición 1: MACD semanal > 0 y decreciente 

TIMEFRAME(weekly) 

MiMACD = MACD[12,26,9](Close) 

c1 = MiMACD > 0 AND MiMACD < MiMACD[1] 

REM Condición 2: Estocástico diario > 70 

TIMEFRAME(daily) 

MiSTO = Stochastic[14,3](Close) 

c2 = MiSTO > 70 

REM Nivel del stop de venta 

MiStop = Low[1] 

REM Criterio: Posición de los precios respecto al STOP 

Criterio = (Close / MiStop -1) * 100 

REM Condición 3: los precios bajo el STOP o a una distancia inferior a 
un +5% 

c3 = Criterio > -5 

SCREENER[c1 AND c2 AND c3](Criterio) 

 

5. Glosario  

A  

Código Implementación Función 

Abs Abs(a) Función matemática "Valor Absoluto"  

AccumDistr AccumDistr(price) 
Designa la Acumulación Distribución 
clásica  

ADX ADX[N] Indicador Average Direccional Index  

ADXR ADXR[N] 
Indicador Average Direccional Index 
Rate  

AND a AND b Operador lógico Y  

AroonDown AroonDown[P] Designa el Aroon Down  

AroonUp AroonUp[P] Designa el Aroon Up  

Atan Atan(a) Función matemática "Arctangente"  

AS RETURN Result AS"ResultName" 
Instrucción utilizada para atribuir un 
nombre a una curva  

Average Average[N](price) Media Móvil Aritmética  

AverageTrueRange AverageTrueRange[N](price) 
Designa la Media móvil con alisado de 
Wilder del True Range  

B  

Código Implementación Función 

BarIndex BarIndex 
Cuenta el número de velas en el 
gráfico  

BollingerBandWidth BollingerBandWidth[N](price) Ancho de banda de Bollinger  

BollingerDown BollingerDown[N](price) Soporte de la banda de Bollinger  

BollingerUp BollingerUp[N](price) Resistencia de la banda de Bollinger  

BREAK 
BREAK 
(FOR...DO...BREAK....NEXT) o 
(WHILE...DO...BREAK...WEND) 

Instrucción de salida forzada de bucle 
FOR o WHILE  

C  

Código Implementación Función 

CALL MyResult = CALL MyFunction  Llamada de función del usuario 

CCI 
CCI[N](price) o CCI[N] aplicado 
por defecto en TypicalPrice 

Commodity Channel Index  

ChaikinOsc ChaikinOsc[Ch1, Ch2](price) Designa el oscilador de Chaikin  

Chandle Chandle[N](price) 
Designa el Chande Momentum 
Oscillator  

ChandeKrollStopUp ChandeKrollStopUp[Pp, Qq, X] 
Stop de protección según Chande y 
Kroll en posición compradora  

ChandeKrollStopDown ChandeKrollStopDown[Pp, Qq, X] 
Stop de protección según Chande y 
Kroll en posición vendedora 

Close Close[N] 
Designa el precio de cierre de la vela 
actual o de n días previos 

COLOURED RETURN COLOURED(R,G,B) 
Result Colorea una curva de un cierto 
color según la convención RGB  

COS COS(a) Función Coseno  

Crosses Over a Crosses Over b 
Operador booleano que verifica que 
una curva pase por encima de otra  

Crosses Under a Crosses Under b 
Operador booleano que verifica que 
una curva pase por debajo de otra  

CUMSUM CUMSUM(price) 
Sumatorio de un precio desde el inicio 
del histórico mostrado  

CurrentDayOfWeek CurrentDayOfWeek Designa el día actual  

CurrentHour CurrentHour Designa la hora actual  

CurrentMinute CurrentMinute Designa el minuto actual  

CurrentMonth CurrentMonth Designa el mes actual  

CurrentSecond CurrentSecond Designa el segundo actual  

CurrentTime CurrentTime Designa HoraMinuto actual  

CurrentYear CurrentYear Designa el año actual  

CustomClose CustomClose[N]  
Constante configurable en la ventana 
de propiedades al mostrar el gráfico 
(por defecto, cierre)  

Cycle Cycle(price) Indicador Ciclo (en precio)  

D  

CÓDIGO IMPLEMENTACIÓN FUNCIÓN  

Date Date[N] 
Designa la fecha de cierre de la vela 
actual  

Day Day[N] Día de cierre de la vela actual  

Days Days[N] Contador de días desde 1900  

DayOfWeek DayOfWeek[N] 
Designa el día de la semana durante el 
cual la vela actual se ha cerrado  

Dclose Dclose(N) 
Precio de cierre del enésimo día previo 
al de la vela actual  

DEMA DEMA[N](price) Doble Media Móvil Exponencial  

Dhigh Dhigh(N) 
Precio máximo del enésimo día previo 
al de la vela actual  

DI DI[N](price) 
Designa el Demand Index (Índice de la 
Demanda)  

DIminus Diminus[N](price) Designa el DI- 

Diplus Diplus[N](price) Designa el DI+  

Dlow Dlow(N) 
Precio mínimo del enésimo día previo 
al de la vela actual  

DO Ver FOR et WHILE 
Instrucción opcional de los bucles FOR 
y WHILE para introducir la acción de 
inicio de bucle  

Dopen Dopen(N) 
Precio de apertura del enésimo día 
previo al de la vela actual  

DOWNTO Ver FOR 
Instrucción que se aplica sobre el bucle 
FOR para ordenar una lectura 
decreciente  

DPO DPO[N](price) Designa el Detrented Price Oscillator  

E  

Código Implementación Función 

EaseOfMovement EaseOfMovement[I] Designa el indicador Ease of Movement  

ELSE Ver IF/THEN/ELSE/ENDIF 
Instrucción de llamada de la segunda 
condición a defecto de la primera salida 
de IF  

ELSEIF Ver IF/THEN/ELSE/ENDIF 

Contracción de ELSE IF (imbricación en 
un conjunto de instrucciones 
condicionales de otro conjunto de 
instrucciones)  

EMV EMV[N] 
Designa el indicador Ease of Movement 
Value  

EQUITYFRAME EQUITYFRAME("marché", "ticker") 
Busca una condición en relación con un 
valor del mismo mercado  

ENDIF Ver IF/THEN/ELSE/ENDIF 
Instrucción a introducir al final del 
conjunto de instrucciones condicionales  

EndPointAverage EndPointAverage[N](price) Ultimo punto de Media Móvil  

EstimatedVolume EstimatedVolume 
Indica una estimación anticipada del 
volumen  

Exp Exp(a) Función matemática "Exponencial"  

ExponentialAverage ExponentialAverage[N] (price) Media Móvil Exponencial  

 

F-G  

Código Implementación Función 

FOR/TO/NEXT FOR i=a TO b DO a NEXT 
Bucle "Para" (Toma todos los valores 
del inicio al final designado, o vice 
versa (DOWNTO))  

ForceIndex ForceIndex(price) 
Da el indicador Force Index, que 
determina quién controla el mercado 
(vendedor, comprador)  

 

H 

Código Implementación Función  

High High[N] 
Designa el precio máximo alcanzado 
durante el período N  

Highest Highest[N](price) 
Designa el máximo de un período sur 
un horizonte temporal dado  

HistoricVolatility HistoricVolatility[N](price) 
Designa la volatilidad histórica (o 
volatilidad estadística)  

Hour Hour[N] 
Designa la hora de cierre de cada vela 
del histórico  

I-J-K 

Código Implementación Función 

IF/THEN/ENDIF IF a THEN b ENDIF 
Conjunto de instrucciones 
condicionales sin segunda condición  

IF/THEN/ELSE/ENDIF IF a THEN b ELSE c ENDIF 
Conjunto de instrucciones 
condicionales  

IntradayBarIndex IntradayBarIndex[N] 
Cuenta el número de velas en el 
gráfico intraday  

 

L 

Código Implementación Función 

LinearRegression LinearRegression[N](price) Regresión lineal  

LinearRegressionSlope LinearRegressionSlope[N] Pendiente de la regresión lineal (price)  

Log (logarithm) Log(a) 
Función matemática "logaritmo 
neperiano"  

Low Low[N] 
Designa el mínimo alcanzado durante 
el período  

Lowest Lowest[N](price) 
Designa el mínimo de un período 
dentro de un horizonte temporal dado  

 

M 

Código Implementación Función 

MACD MACD[S,L,Si](price) 
Designa al Moving Average 
Convergence Divergence (MACD) en 
histograma  

MACDline MACDLine[S,L](price) Designa la línea del MACD  

MassIndex MassIndex[N] 
Indicador Mass Index aplicado en N 
velas  

Max Max(a,b) Función matemática "Máximo"  

MedianPrice MedianPrice Media del precio máximo y del mínimo  

Min Min(a,b) Función matemática "Mínimo"  

Minute Minute 
Designa el minuto del instante del 
cierre de cada vela del histórico  

Mod a Mod b 
Función matemática "Resto del 
cociente euclídeo"  

Momentum Momentum[I]  
Designa el Momentum (precio de cierre 
actual – precio de cierre de la enésima 
vela precedente)  

MoneyFlow MoneyFlow[N](price) Designa el MoneyFlow entre -1 y 1  

MoneyFlowIndex MoneyFlowIndex[N] Designa el MoneyFlowIndex  

Month Month[N] 
Designa el mes de cierre de cada vela 
del histórico  

 



 

 

 © Bankinter, S.A. Todos los derechos reservados Página 30

Broker Plus 

Guía de programación ProScreenerTM 

N 

Código Implementación Función 

NEXT Ver FOR/TO/NEXT 
Instrucción a introducir al final del 
bucle "Para" (FOR)  

NOT NOT a Operador lógico NO  

 

O 

Código Implementación Función 

OBV OBV(price) Designa el "On-Balance-Volume"  

ONCE 
ONCE VariableName = 
VariableValue 

Instrucción que precede a otra que 
deseamos ejecutar una sola vez  

Open Open[N 
Designa el precio de apertura de la 
vela actual o celle de n días previos  

OpenOfNextBar OpenOfNextBar 
Designa el precio de apertura de la 
vela siguiente  

OR a OR b Operador lógico O  

 

P-Q 

Código Implementación Función 

PriceOscillator PriceOscillator[S,L](price) Indicador Percentage Price oscillator  

PositiveVolumeIndex PriceVolumeIndex(price) 
Designa el indicador Positive Volume 
Index  

PVT PVT(price) 
Designa el indicador "Price Volume 
Trend"  

R 

Código Implementación Función 

R2 R2[N](price) 
Designa el coeficiente R Cuadrado 
(error de precios en la regresión linear)  

Range Range[N] 
Devuelve el Range (rango, diferencia 
entre el precio máximo y mínimo de la 
vela actual)  

REM REM comentario 
Precede un comentario (el código no lo 
toma en cuenta, pero facilitan la 
lectura al usuario)  

Repulse Repulse[N](price) 
Devuelve el indicador Repulse (mide la 
fuerza alcista y bajista de cada vela)  

RETURN RETURN 
Resultado Instrucción que envía el 
resultado  

ROC ROC[N](price) Designa el "Price Rate of Change"  

RSI RSI[N](price) 
Designa el oscilador "Relative Strength 
Index"  

Round Round(a) 
Función matemática "Redondeo a la 
unidad" (parte entera)  

 

S 

Código Implementación Función 

SAR SAR[At,St,Lim] Designa el Parabólico SAR  

SARatdmf SARatdmf[At,St,Lim](price) 
Designa el Parabólico SAR en el módulo 
ATDMF  

SCREENER SCREENER[c](price) Muestra los resultados  

Sin Sin(a) Función matemática "Seno"  

Sgn Sgn(a) Función matemática "Signo de"  

SMI SMI[N,SS,DS](price) 
Designa el índice Estocástico 
Momentum (Stochastic Momentum 
Index)  

SmoothedStochastic SmoothedStochastic[N,K] (price) Designa un estocástico alisado  

SORT BY Screener(c1) SORT BY price 
Permite visualizar u ordenar los 
resultados  

Square Square(a) 
Función matemática "Cuadrado" 
(potencia 2)  

Sqrt Sqrt(a) Función matemática "Raíz cuadrada"  

STD STD[N](price) Función estadística "Desviación Típica"  

STE STE[N](price) Función estadística "Error típico"  

Stochastic Stochastic[N,K](price) Línea %K del Estocástico  

Summation Summation[N](price) 
Suma de un cierto precio de las N 
últimas velas  

SuperTrend SuperTrend[STF,N] Designa el Super Trend  

T 

Código Implementación Función 

Tan Tan(a) Función matemática "Tangente"  

TEMA TEMA[N](price) Media Móvil Exponencial Triple  

THEN Ver IF/THEN/ELSE/ENDIF 
Instrucción que sigue la primera 
condición del conjunto condicional "IF"  

Time Time[N] 
Da la evolución de la 
HoraMinutoSegundo y permite llamar 
la hora en el programa  

TimeSeriesAverage TimeSeriesAverage[N](price) Media Móvil de las series temporales  

TIMEFRAME TIMEFRAME(P) 
Permite trabajar en un período de 
tiempo distinto del período por defecto  

TO Ver FOR/TO/NEXT 
Instrucción "hasta " en el bucle "Para" 
(FOR)  

Today Today Fecha actual  

TotalPrice TotalPrice[N] 
(cierre + apertura + máximo + 
mínimo) / 4  

TR TR(price) Designa el True Range  

TriangularAverage TriangularAverage[N](price) Media Móvil Triangular  

TRIX TRIX[N](price) Triple Media Móvil Exponencial  

TypicalPrice TypicalPrice[N] 
Designa el precio Típico (Media de 
máximo, mínimo y cierre)  

 

U 

Código Implementación Función 

Undefined  a = Undefined 
Permite dejar una variable indefinida 
(es un tipo de variable) 

 

V 

Código Implementación Función 

Variation Variation(price) 
Da la diferencia entre el cierre de la 
víspera y el cierre actual en % 

Volatility Volatility[S, L] Designa la volatilidad de Chaikin 

Volume Volume[N] Designa el volumen 

VolumeOscillator VolumeOscillator[S,L] Designa el oscilador de volumen 

VolumeROC VolumeROC[N] 
Designa el volumen del Rate Of Change 
(ROC) 

W 

Código Implementación Función 

WeightedAverage WeightedAverage[N](price) Designa la Media Móvil Ponderada  

WeightedClose WeightedClose[N] 
Da la Media entre el precio de cierre, 
máximo y mínimo, con ponderaciones 
respectivas de 2, 1 y 1  

WEND Ver WHILE/DO/WEND 
Instrucción a introducir al final del 
bucle While/Do/Whend (mientras)  

WHILE/DO/WEND 
WHILE (condición) DO (acción) 
WEND 

Bucle "Mientras"  

WilderAverage WilderAverage[N](price) Da la Media Móvil de Wilder  

Williams Williams[N](close) Calcula el %R de Williams  

WilliamsAccumDistr WilliamsAccumDistr(price) 
Indicador Acumulación/Distribución de 
Williams  

 

X 

Código Implementación Función 

XOR a XOR b Operador lógico O exclusivo  
 

Y 

Código Implementación Función  

Year Year[N] 
Da la evolución de los años y permite 
llamar a los años en el programa  

Yesterday Yesterday[N] 
Da la evolución del día de ayer y 
permite llamar a este último en el 
programa  

 

Z 

Código Implementación Función 

ZigZag ZigZag[Zr](price) 
Designa los Zig-Zag de la teoría de las 
ondas de Eliott 

ZigZagPoint ZigZagPoint[Zp](price) 
Designa los Zig-Zag de la teoría de las 
ondas de Eliott calculadas a Z puntos 

OTROS 

Código Función  

+ Operador de adición  

- Operador de sustracción  

* Operador de multiplicación  

/ Operador de división decimal  

= Operador de igualdad  

<> Operador de diferencia  

< Operador de inferioridad estricta  

> Operador de superioridad estricta  

<= Operador de inferioridad  

>= Operador de superioridad 

